# Programming and Proving in Coq

#### Maria João Frade

HASLab - INESC TEC Departamento de Informática, Universidade do Minho

2020/2021

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

VF 2020/21 1 / 34

## Some datatypes of programming

```
Inductive unit : Set := tt : unit.
Inductive bool : Set := true : bool | false : bool.
Inductive nat : Set := 0 : nat | S : nat -> nat.
Inductive option (A : Type) : Type := Some : A -> option A
                                    | None : option A.
```

## Roadmap

#### **Programming and Proving in Coq**

- some datatypes of programming;
- functional correctness; partiality; specification types;
- program extraction;
- non-structural recursion.

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

VF 2020/21 2 / 34

Some operations on bool are also provided: andb (with infix notation &&), orb (with infix notation ||), xorb, implb and negb.

## Some datatypes of programming

```
Inductive sum (A B : Type) : Type := inl : A -> A + B
                                   | inr : B -> A + B.
Inductive prod (A B : Type) : Type := pair : A -> B -> A * B.
Definition fst (A B : Type) (p : A * B) := let (x, _) := p in x.
Definition and (A B : Type) (p : A * B) := let (, y) := p in y.
```

The constructive sum  $\{A\}+\{B\}$  of two propositions A and B.

```
Inductive sumbool (A B : Prop) : Set :=
  | left : A \rightarrow {A} + {B}
  | right : B -> {A} + {B}.
```

Maria João Frade (HASLab, DI-UM)

VF 2020/21 4 / 34

## If-then-else

- The sumbool type can be used to define an "if-then-else" construct in Cog.
- Cog accepts the syntax if test then ... else ... when test has either of type bool or  $\{A\}+\{B\}$ , with propositions A and B.
- Its meaning is the pattern-matching match test with left H => ... right H => ... end.
- We can identify {P}+{~P} as the type of decidable predicates:

```
The standard library defines many useful predicates, e.g.
le_lt_dec : forall n m : nat, {n <= m} + {m < n}
Z.eq_dec : forall x y : Z, \{x = y\} + \{x \leftrightarrow y\}
Z.lt_ge_dec : forall x y : Z, \{x < y\} + \{x >= y\}
```

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

VF 2020/21 5 / 34

## If-then-else

```
A function that checks if an element is in a list.
Fixpoint elem (a:Z) (1:list Z) {struct 1} : bool :=
  match 1 with
    | nil => false
    | cons x xs => if (Z.eq_dec x a) then true else (elem a xs)
  end.
```

```
Exercise:
Inspect the proof of
Proposition elem_corr : forall (a:Z) (11 12:list Z),
                         elem a (app 11 12) = orb (elem a 11) (elem a 12).
and prove the following lemma:
Lemma ex : forall (a:Z) (11 12:list Z),
                         elem a (app 11 (cons a 12)) = true.
```

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

VF 2020/21 7 / 34

#### Exercises

Load the file lesson3.v in the Cog proof assistant to follow the examples of the coming slides. Analyse the examples and solve the exercises proposed.

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

VF 2020/21 6 / 34

## The "subset" type

 Cog's type system allows to combine a datatype and a predicate over this type, creating "the type of data that satisfies the predicate". Intuitively, the type one obtains represents a subset of the initial type.

```
Inductive sig (A : Type) (P : A -> Prop) : Type :=
    exist : forall x : A, P x \rightarrow sig A P.
```

- Given A: Type and P: A->Prop, the syntactical convention for (sig A P) is the construct  $\{x:A \mid P \mid x\}$ . (Predicate P is the *characteristic function* of this set).
- We may build elements of this set as (exist x p) whenever we have a witness x:A with its justification p:(P x).
- From such a (exist x p) we may in turn extract its witness x:A.
- In technical terms, one says that sig is a "dependent sum" or a  $\Sigma$ -type.

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

VF 2020/21 8 / 34

## The "subset" type

A value of type  $\{x:A \mid P \mid x\}$  should contain a *computation component* that says how to obtain a value v and a *certificate*, a proof that v satisfies predicate P.

A variant sig2 with two predicates is also provided.

```
Inductive sig2 (A : Type) (P Q : A -> Prop) : Type :=
    exist2 : forall x : A, P x \rightarrow Q x \rightarrow sig2 A P Q
```

The notation for (sig2 A P Q) is  $\{x:A \mid P \times \& Q \times \}$ .

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

VF 2020/21 9 / 34

### **Partiality**

The Cog system does not allow the definition of partial functions (i.e. functions that give a run-time error on certain inputs). However we can enrich the function domain with a precondition that assures that invalid inputs are excluded.

- ullet A partial function from type A to type B can be described with a type of the form  $\forall x: A, Px \rightarrow B$ , where P is a predicate that describes the function's domain
- Applying a function of this type requires two arguments: a term t of type A and a proof of the precondition Pt.

## **Functional correctness**

There are two approaches to define functions and provide proofs that they satisfy a given specification:

• To define these functions with a weak specification and then add companion lemmas.

For instance, we define a function  $f: A \rightarrow B$  and we prove a statement of the form  $\forall x: A, Rx(fx)$ , where R is a relation coding the intended input/output behaviour of the function.

• To give a *strong specification* of the function: the type of this function directly states that the input is a value x of type A and that the output is the combination of a value v of type B and a proof that v satisfies Rxv.

This kind of specification usually relies on dependent types.

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

VF 2020/21 10 / 34

# Example: the function head

## An attempt to define the head function as follows will fail!

```
Definition head (A:Type) (1:list A) : A :=
  match 1 with
  | cons x xs => x
  end.
Error: Non exhaustive pattern-matching: no clause found
       for pattern nil
```

To overcome the above difficulty, we need to:

- consider a precondition that excludes all the erroneous argument values:
- pass to the function an additional argument: a proof that the precondition holds:
- the match constructor return type is lifted to a function from a proof of the precondition to the result type.
- any invalid branch in the match constructor leads to a logical contradiction (it violates the precondition).

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

VF 2020/21

# Example: the function head

```
Definition head (A:Type) (1:list A) : 1<>nil -> A.
refine (
  match 1 as 1' return 1'<>nil -> A with
 | nil => fun H =>
  | cons x xs => fun H => x
  end ).
elimtype False; apply H; reflexivity.
Defined.
```

```
Print Implicit head.
head : forall (A : Type) (1 : list A), 1 <> nil -> A
Arguments A, 1 are implicit
```

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

VF 2020/21 13 / 34

#### Extraction

- Conventional programming languages do not provide dependent types and well-typed functions in Cog do not always correspond to well-typed functions in the target programing language.
- In CIC functions may contain subterms corresponding to proofs that have practically no interest with respect to the final value.
- The computations done in the proofs correspond to verifications that should be done once and for all at compile-time, while the computation on the actual data needs to be done for each value presented to functions at run-time.
- Coq implements this mechanism of filtering the computational content from the objects - the so called extraction mechanism.
- The distinction between the sorts Prop and Set is used to mark the logical aspects that should be discharged during extraction or the computational aspects that should be kept.

## Example: the function head

The specification of head is:

```
Definition headPre (A:Type) (1:list A) : Prop := 1<>nil.
Inductive headRel (A:Type) (x:A) : list A -> Prop :=
    headIntro : forall 1. headRel x (cons x 1).
```

The correctness of function head is thus given by the following lemma:

```
Lemma head_correct : forall (A:Type) (1:list A) (p:headPre 1),
                     headRel (head p) 1.
Proof.
  destruct 1.
 - intro H; elim H; reflexivity.
  - intros; destruct 1; [simpl; constructor | simpl; constructor]
Qed.
```

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

VF 2020/21 14 / 34

#### Extraction

Coq supports different target languages: Ocaml, Haskell, Scheme. The extraction framework must be loaded explicitly.

```
Require Extraction.
```

```
Check head.
head : forall (A : Type) (1 : list A), 1 \Leftrightarrow nil \rightarrow A
```

```
Extraction Language Haskell.
Extraction Inline False rect.
Extraction head.
```

```
head :: (List a1) -> a1
head 1 =
  case 1 of
   Nil -> Prelude.error "absurd case"
    Cons x xs -> x
```

Maria João Frade (HASLab. DI-UM)

Programming and Proving in Coq

#### Extraction

Extraction of all the mentioned objects and all their dependencies in the Cog toplevel.

```
Recursive Extraction head.
module Main where
import qualified Prelude
data List a =
   Nil
 | Cons a (List a)
head :: (List a1) -> a1
head 1 =
  case 1 of {
  Nil -> Prelude.error "absurd case";
   Cons x - \rightarrow x
```

Recursive extraction of all the mentioned objects and all their dependencies into a file.

```
Extraction "filename" head.
Maria João Frade (HASLab, DI-UM)
                                 Programming and Proving in Coq
                                                                           VF 2020/21 17 / 34
```

## Specification types

Using  $\Sigma$ -types we can express specification constrains in the type of a function we simply restrict the codomain type to those values satisfying the specification.

• Consider the following definition of the inductive relation "x is the last element of list I", and the theorem specifing the function that gives the last element of a list.

```
Inductive Last (A:Type) (x:A) : list A -> Prop :=
| last_base : Last x (x :: nil)
| last_step : forall 1 y, Last x 1 -> Last x (y :: 1).
```

```
Theorem last_correct : forall (A:Type) (1:list A),
                                   1<>nil -> { x:A | Last x 1 }.
```

- By proving this theorem we build an inhabitant of this type, and then we can extract the computational content of this proof, and obtain a function that satisfies the specification.
- The Cog system thus provides a certified software production tool, since the extracted programs satisfy the specifications described in the formal developments.

Maria João Frade (HASLab, DI-UM) Programming and Proving in Coq VF 2020/21 19 / 34

## Extraction

The system also provides a mechanism to specify terms for inductive types and constructors of the target programming language.

For instance, we may want to use the Haskell native list type instead of the Coq one.

```
Extract Inductive list => "[]" [ "[]" "(:)" ].
```

```
Recursive Extraction head.
module Main where
import qualified Prelude
head :: ([] a1) -> a1
head 1 =
  case 1 of {
  [] -> Prelude.error "absurd case";
  (:) x -> x
```

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

VF 2020/21 18 / 34

## Specification types

Let us build an inhabitant of that type

```
Theorem last_correct : forall (A:Type) (1:list A),
                                 1 <> nil -> \{ x:A \mid Last x l \}.
Proof.
  induction 1.
  - intro H; elim H; reflexivity.
  - intros. destruct 1.
    + exists a. constructor.
    + elim IH1.
      * intros; exists x. constructor. assumption.
      * discriminate.
Qed.
```

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

# Program extraction

We can extract the computational content of the proof of the last theorem.

```
Extraction Inline False rect.
Extraction Inline sig_rect.
Extraction Inline list rect.
Recursive Extraction last correct.
module Main where
import qualified Prelude
type Sig a = a
  -- singleton inductive, whose constructor was exist
last_correct :: ([] a1) -> a1
last correct 1 =
  case 1 of {
   [] -> Prelude.error "absurd case":
   (:) y 10 -> case 10 of {
                 [] -> y;
                 (:) _ _ -> last_correct 10}}
Maria João Frade (HASLab, DI-UM) Programming and Proving in Coq
                                                              VF 2020/21 21 / 34
```

## Case study: sorting a list

A simple characterisation of sorted lists consists in requiring that two consecutive elements be compatible with the < relation.

We can codify this with the following predicate:

```
Open Scope Z_scope.
Inductive Sorted : list Z -> Prop :=
  | sorted0 : Sorted nil
  | sorted1 : forall z:Z, Sorted (z :: nil)
  | sorted2 : forall (z1 z2:Z) (l:list Z),
        z1 \le z2 -> Sorted (z2 :: 1) -> Sorted (z1 :: z2 :: 1).
```

#### Exercise

#### Exercise

Built an alternative definition of function head called "head\_corr" based on the strong specification mechanism provided by Cog.

That is,

- give a strong specification of "head\_corr";
- prove it:
- and then, extract the computational content of this proof.

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

VF 2020/21 22 / 34

## Case study: sorting a list

To capture permutations, instead of an inductive definition we will define the relation using an auxiliary function that count the number of occurrences of elements:

```
Fixpoint count (z:Z) (1:list Z) {struct 1} : nat :=
 match 1 with
  I nil => 0%nat
  | (z' :: 1') =>
      match Z.eq_dec z z' with
      | left _ => S (count z 1')
      | right _ => count z l'
      end
  end.
```

A list is a permutation of another when contains exactly the same number of occurrences (for each possible element):

```
Definition Perm (11 12:list Z) : Prop :=
                 forall z, count z 11 = count z 12.
```

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

VF 2020/21

## Case study: sorting a list

#### Exercise:

Prove that Perm is an equivalence relation:

```
Lemma Perm_reflex : forall 1:list Z, Perm 1 1.
Lemma Perm_sym : forall 11 12, Perm 11 12 -> Perm 12 11.
```

Lemma Perm\_trans : forall 11 12 13,

Perm 11 12 -> Perm 12 13 -> Perm 11 13.

#### Exercise:

Prove the following lemmas:

```
Lemma Perm_cons : forall a 11 12,
```

Perm 11 12 -> Perm (a::11) (a::12).

Lemma Perm\_cons\_cons : forall x y 1, Perm (x::y::1) (y::x::1).

Maria João Frade (HASLab, DI-UM) Programming and Proving in Coq

VF 2020/21 25 / 34

### Case study: sorting a list

The theorem we want to prove is:

```
Theorem isort_correct : forall (1 1':list Z),
                       l'=isort l -> Perm l l' /\ Sorted l'.
```

We will certainly need auxiliary lemmas... Let us make a prospective proof attempt:

#### Proof.

```
induction 1; intros.
- unfold Perm; rewrite H; split; auto. simpl. constructor.
- simpl in H.
  rewrite H. (* ????????? *)
```

```
a : Z
IHl : forall 1' : list Z, 1' = isort 1 -> Perm 1 1' /\ Sorted 1'
l': list Z
H: 1' = insert a (isort 1)
______
Perm (a :: 1) (insert a (isort 1)) /\ Sorted (insert a (isort 1))
```

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

VF 2020/21 27 / 34

## Case study: sorting a list

A simple strategy to sort a list consist in iterate an "insert" function that inserts an element in a sorted list.

```
Fixpoint insert (x:Z) (1:list Z) {struct 1} : list Z :=
 match 1 with
 | nil => x :: nil
  | (h :: t) =>
        match Z.lt_ge_dec x h with
         left _ => x :: (h :: t)
        | right _ => h :: (insert x t)
        end
  end.
```

```
Fixpoint isort (1:list Z) : list Z :=
 match 1 with
   nil => nil
  | (h :: t) => insert h (isort t)
```

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

VF 2020/21 26 / 34

# Case study: sorting a list

It is now clear what are the needed lemmas:

```
Lemma insert_Perm : forall x 1, Perm (x::1) (insert x 1).
Lemma insert Sorted: forall x 1. Sorted 1 -> Sorted (insert x 1).
```

In order to prove them the following lemmas about count, may be useful.

```
Lemma count_insert_eq : forall x 1,
                            count x (insert x 1) = S (count x 1).
Lemma count_cons_diff : forall z x 1,
                            z \leftrightarrow x \rightarrow count z 1 = count z (x :: 1).
Lemma count_insert_diff : forall z x 1,
                             z \leftrightarrow x \rightarrow count z l = count z (insert x l).
```

Check the proofs.

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

## Case study: sorting a list

Now we can conclude the proof of correctness...

```
Theorem isort_correct : forall (1 1':list Z),
                       l'=isort l -> Perm l l' /\ Sorted l'.
Proof.
  induction 1; intros.
  - unfold Perm; rewrite H; split; auto. simpl. constructor.
  - simpl in H.
    rewrite H. (* ????????? *)
    elim (IHl (isort 1)); intros; split.
    + apply Perm_trans with (a::isort 1).
      * unfold Perm. intro z. simpl. elim (Z.eq_dec z a).
        -- intros. elim HO; reflexivity.
        -- auto with zarith.
      * apply insert Perm.
    + apply insert_Sorted. assumption.
Qed.
```

Maria João Frade (HASLab, DI-UM) Programming and Proving in Coq

VF 2020/21 29 / 34

#### Non-structural recursion

When the recursion pattern of a function is not structural in the arguments, we are no longer able to directly use the derived recursors to define it.

#### Consider the Euclidean division algorithm written in Haskell

```
div :: Int -> Int -> (Int.Int)
div n d | n < d = (0,n)
        | otherwise = let (q,r) = div (n-d) d
                      in (q+1,r)
```

- In recent versions of Coq (after v8.1), a new command Function allows to directly encode general recursive functions.
- The Function command accepts a measure function that specifies how the argument "decreases" between recursive function calls.
- It generates proof-obligations that must be checked to guaranty the termination.

## Case study: sorting a list

#### Exercise:

```
Complete the following proof and extract its computational content to an Haskell
function.
```

```
Definition inssort : forall (1:list Z),
                              { 1' | Perm 1 1' & Sorted 1' }.
 induction 1.
 - exists nil. constructor. constructor.
 - elim IHl. intros. exists (insert a x).
Defined.
```

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

VF 2020/21 30 / 34

#### Non-structural recursion

```
Close Scope Z_scope.
Require Import Recdef. (* because of Function *)
Function div (p:nat*nat) {measure fst} : nat*nat :=
 match p with
 |(_,0)| => (0,0)
  | (a,b) =  if le_lt_dec b a
            then let (x,y) := div (a-b,b) in (1+x,y)
             else (0.a)
  end.
Proof.
intros. simpl. omega.
Qed.
```

The Function command generates a lot of auxiliary results related to the defined function. Some of them are powerful tools to reason about it.

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

VF 2020/21 32 / 34

## Non-structural recursion

The Function command is also useful to provide "natural encodings" of functions that otherwise would need to be expressed in a contrived manner.

#### Exercise:

Complete the definition of the function merge, presenting a proof of its termination.

```
Function merge (p:list Z*list Z)
{measure (fun p=>(length (fst p))+(length (snd p)))} : list Z :=
 match p with
  | (nil,1) => 1
  | (1,nil) => 1
  | (x::xs,y::ys) => if Z.lt_ge_dec x y
                     then x::(merge (xs,y::ys))
                     else y::(merge (x::xs,ys))
  end.
```

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

VF 2020/21 33 / 34

# Another example of correctness

```
A specification of the Euclidean division algorithm:
Definition divRel (args:nat*nat) (res:nat*nat) : Prop :=
           let (n,d):=args in let (q,r):=res in q*d+r=n / r< d.
Definition divPre (args:nat*nat) : Prop := (snd args)<>0.
```

```
A proof of correctness:
Theorem div_correct : forall (p:nat*nat), divPre p -> divRel p (div p).
 (* we make use of the specialised induction principle to conduct the proof... *)
 functional induction (div p); simpl.
 - intro H; elim H; reflexivity.
 - (* a first trick: we expand (div (a-b,b)) in order to get rid of the let (q,r)=... *)
   replace (div (a-b,b)) with (fst (div (a-b,b)), snd (div (a-b,b))) in IHpO.
    + simpl in *. intro H; elim (IHpO H); intros. split.
      * (* again a similar trick: we expand "x" and "y0" in order to use an hypothesis *)
        change (b + (fst (x,y0)) * b + (snd (x,y0)) = a).
       rewrite <- e1. omega.
      * (* and again... *)
       change (snd (x,y0)<b); rewrite <- e1; assumption.
    + symmetry; apply surjective_pairing.
```

Maria João Frade (HASLab, DI-UM)

Programming and Proving in Coq

VF 2020/21 34 / 34